

## 連載

## Mitaka 改造の手引き (6) 最終回

## ～字幕と地名：SPACE-CISTE 簡易版～

河村聡人 (Mitaka ワーキンググループ)

## 1. はじめに

Mitaka (©2005-2023 加藤恒彦, 国立天文台 4次元デジタル宇宙プロジェクト) のカスタマイズ機能[1]の活用と普及のための連載 (全6回) は今回で最終回となります。

連載の各回の副題

1. 保存と読み込み：日食 (一地点) [2]
2. コマンドセット 1：日食 (多地点) [3]
3. コマンドセット 2：星座[4]
4. シーケンス 1：月の満ち欠け (宇宙) [5]
5. シーケンス 2：月の満ち欠けと日食 [6]
6. 字幕と地名：SPACE-CISTE 簡易版

当連載は Mitaka ワーキンググループ

(WG) による講習 (ワークショップ) [7]

向け教材「Mitaka カスタマイズ攻略マニュアル」を下書きとしています。この他にも多数のワークショップ向け教材が、次の QR コード先にあります。これら教材や当連載は Mitaka 同梱の公式マニュアル[8]を補完・要約するものです。



図1 MitakaWG の Google ドライブ[9]

当連載で扱う内容は Mitaka の利用上の注意[10]に従い、自由な再配布が認められた範囲[11]です。なお Mitaka のスクリーンショットの掲載には別途許可を取っています。

今回紹介する機能は字幕と地名です。その応用例として宇宙お宝探しゲーム

SPACE-CISTE があります[12][13][14]。SPACE-CISTE はクイズに示された地点を探し宇宙を探索する Mitaka 上で動くゲームで、Mitaka 初心者向けの操作の習熟と宇宙の知識を深める事を目的としています。本記事では、字幕の機能紹介を行いつつ、その応用問題として SPACE-CISTE の簡易版に取り組みます。

## 2. 開発環境の確認

Mitaka プラグイン開発のための環境については第1回 (Mitaka の実行環境の説明) 及び第2回 (エディタに関する説明) を参考にしてください。

## 3. 字幕

Mitaka には画面上に字幕を表示する機能があり、使用者に必要な情報を提供することができます。字幕を表示させるには以下のように SUBTITLES\_DISP のスイッチキーをオンにしてください。

```
SwitchOn(SUBTITLES_DISP)
```

字幕の位置には次の3つの基準があり、今回は1番目と3番目を紹介します。

- ・ ウインドウ基準
- ・ 標準の正面方向基準 (ドーム投影や VR 表示用)
- ・ オブジェクトのラベル表示位置基準 (天体等に付随するもの)

## 3.1 ウインドウ基準の字幕

Mitaka に表示されるロゴやクレジットのように、視線の移動などに影響されず、ウイ

ンドウ上の特定の位置に常に表示される字幕です。ウインドウの縦幅は、そのサイズによらず、字幕 24 行分とされます。横の文字数はウインドウの縦横比率によって変化します。例えば 1920×1020 ピクセルのディスプレイで最大化した場合なら、横幅は全角 47 文字幅になります。全画面表示にするとメニューバーの分だけ縦長に比率が変わりますので、横幅の文字数は少なくなります。

字幕は一行ずつ別のコマンドとして書く必要があります。例えば次の 4 コマンドはウインドウ左側上部にキー操作のガイドとして「メニューを開く」等とそのキーを表示します (図 2 及び図 3 の左側上部)。

```
SetSubtitle (WindowKey:L1,
  Text:"メニューを開く/選ぶ:",
  Color:#10FFFF, X:-16.75, Y:18.6)
SetSubtitle (WindowKey:L2,
  Text:"・X キー",
  Color:#10FFFF, X:-19.5, Y:17.3)
SetSubtitle (WindowKey:L3,
  Text:"メニューを閉じる:",
  Color:#10FFFF, X:-17.5, Y:14.9)
SetSubtitle (WindowKey:L4,
  Text:"・Z キー",
  Color:#10FFFF, X:-19.5, Y:13.6)
```

各パラメータの詳細は以下となります。

**WindowKey:** 各字幕固有の識別子です。

**Text:** 表示する文字列です。「,」まで、もしくは「"」で括られた一連の文字を表示する文字列として認識します。

**X:** ウインドウ中心からみた、字幕の中心の横の位置で、単位は全角 1 文字です。

**Y:** ウインドウ下端からみた、字幕の下端の位置で、単位は 1 文字です。

**Color:** 文字の色で、RGB もしくは RGBA (A は透過率) の、#から始まる 16 進数のカラーコードです。

**Size:** フォントサイズです。デフォルトは 1。

字幕は中央揃えで制御されます。よって左揃えをしたい場合は、字数に応じて位置を調整する必要があります。上記の例では、L1 と L3 の字幕は左揃えとなっているのですが、L1 は全角 10 文字と半角 1 文字なのに対し、L3 は全角 9 文字なので、全角 1.5 文字分の半分、全角 0.75 文字分マイナス方向 (左) に寄っています。なお半角は可変幅フォントのため制御が難しいです。この例の「/」は制御しやすい例外的な半角文字です。左右揃えにおいては、基本的には全角文字を表示するようにしてください。

### 3.2 天体等に付随した字幕

天体名のように、特定の天体等のそばに表示される字幕です。指定した天体等が画面内にあればプラネタリウムモードでも表示されます。宇宙モードでは十分に近づいた場合に表示されます。次の例はすばるに「正」という字幕を表示します (図 2)。

```
SetSubtitle (WindowKey:Ciste_Q1,
  Type:ObjectLabel,
  ObjectKey:HIP_17702,
  Text: "「正」", Color:#10FFFF)
```

ウインドウ基準の字幕と異なり、**Type:ObjectLabel** で天体に付属した字幕であることの宣言と、**ObjectKey:** で字幕を付ける対象の天体などを指定します。他の設定は先述の字幕と共通です。

**ObjectKey:** で天体等を指定する時には **SetTarget** と同様にオブジェクトキーを使用します。星座はオブジェクトキーに含まれません。星座を指定したい場合は、代わりにそれに含まれる恒星をひとつ選んでください。オブジェクトキーについて詳しくは当連載第 2 回の 3.4 章や Mitaka 同梱マニュアル[9]を参考にしてください。




図2 すばるに付随した字幕「正」(上) 宇宙モード、(下) プラネタリウムモード (Mitaka: ©2005-2023 加藤恒彦, 国立天文台 4次元デジタル宇宙プロジェクト)

### 3.3 字幕表示の切り替え

字幕の表示・非表示を切り替える方法は複数あります。場面によって使い分けてください。

字幕の情報を設定も含めて消す場合は、`DeleteSubtitleWindows` にて消す字幕の `WindowKey` を列挙します。たとえば L1~L4 の字幕を一括して消す場合は以下となります。

```
DeleteSubtitleWindows (L1, L2, L3, L4)
```

一時的に字幕を消し、その後色などの設定を再度使用する場合は、表示させる文字列を指定しないようにします。

```
SetSubtitle(WindowKey:Ciste_Q1, Text:)
```

色などの設定は変更せずに、異なる文字を表示する場合は、上記と同じコマンドで `Text:` 以下に表示させたい文字列を指定します。

ほかにも、個別の字幕の表示 (True) / 非表示 (False) を `Display:` パラメータで切り

替えたり、スイッチキー `SUBTITLES_DISP` のオン/オフですべての字幕表示・非表示を切り替える方法もあります。この場合は、字幕を再度表示させるタイミングを逸しないように気を付けてください。

## 4. 地名表示機能

惑星などには天体上に地点を設定し、そこに文字を表示することができます。この機能は地名の追加を本来の目的としていますが、任意の文字列を表示し、先の天体に付随した字幕のように使用できます。この地名表示は宇宙モードでのみ機能し、プラネタリウムモードでは表示されません (図3)。

### 4.1 地点の設定

地点の設定では次のように、地点を識別するための `PlaceKey` とともに対象の天体と天体上の緯度経度だけでなく、表示する文字の書式も設定します。表示する文字そのものは別のコマンドで設定します。この例は月の嵐の大洋に `Ciste_Q2` という地点を設定します。

```
SetPlace(ObjectKey:SAT_E_MOON,
PlaceKey:Ciste_Q2, Lat:18.4,
Lon:-57.4, R1:0.07, Color:#10FFFF,
Size:2)
```

`R1` は字幕を表示する距離を定義するパラメータです。`R1` が大きければ遠くからも、小さくすると近づかないと表示されなくなります。単位は対象の天体の直径で、デフォルト値は 0.1 です。詳細は `features_v174.pdf` [9] に書かれていますが、実際には、見せかたや操作感を考えながら試行錯誤して決めることとなります。

### 4.2 地点に表示する文字

先に設定した地点に対して、表示する文字を設定します。「`PlaceKey:表示する文字列`」の形で、次のように設定します。

### SetStrings(Ciste\_Q2: "「解」")

この例では先ほど設定した `Ciste_Q2` の地点、つまり嵐の大洋に「解」という文字が表示されるようになります (図 3)。




図 3 嵐の大洋に地名として追加された「解」  
(上) 宇宙モードでは表示されるが、(下) プラネタリウムモードでは表示されない  
(Mitaka: ©2005–2023 加藤恒彦, 国立天文台 4次元デジタル宇宙プロジェクト)

## 5. 応用例題 : SPACE-CISTE

本記事の冒頭に紹介したように、SPACE-CISTE は問題に指定された地点にある文字を集めるゲームです。ここでは問題を 2 つ設定してみましょう。

ファイル名は `spaceciste_sample.mcd` とし、次の 4 つのコマンドセットを定義します。

### 1. コマンドセット `SpaceCiste_Init`

問題文字幕 (WindowKey:T1) の初期値として「メニューバーのコマンドから問題を選択してください」を Y: 22.5 に表示。

3.1 章のキーガイドも表示。各問の答えを非表示に変更。

### 2. コマンドセット `SpaceCiste_1`

最初にコマンドセット `SpaceCiste_Init` を呼び出す。問題文「第 1 問 春はあけぼの…星は○○○。」を字幕 T1 に、答え「正」を すばる (HIP\_17702) に表示。

### 3. コマンドセット `SpaceCiste_2`

最初にコマンドセット `SpaceCiste_Init` を呼び出す。問題文「第 2 問 月にある大洋」を字幕 T1 に、答え「解」を嵐の大洋 (月の北緯 18.4 度、西経 57.4 度あたり) に表示。

### 4. コマンドセット `STATE`

コマンド定義ファイル読み込み時に実行される。`SpaceCiste_Init` を呼び出す。

各問のコマンドセットはメニューバーに登録します。文字の色はシアン (#10FFFF) で統一します。この色は地球の一部地域を除き、どこでもある程度目立ちます。

本記事の 3~4 章にて例示したスクリプトはこの応用例題の一部です。完成したスクリプトは末尾の A.1 に掲載しています。

今回の例題は SPACE-CISTE の最低限の骨格となる部分だけを扱っており、改良できる点は多岐にわたります。たとえばキーガイドの充実や基本的な操作を示すチュートリアル、答えやクレジットを表示するシーケンスなどがあります。コマンド定義ファイルの読み込み時にシーケンスを自動的に起動する方法は第 5 回を参考にしてください。これ以降はアイデアと作りこみの世界です。

## 6. おわりに

Mitaka は、バージョン 1.6 より追加されたカスタマイズ機能によって、宇宙を再現するソフトウェアから、より自由な表現・教育・情報共有のプラットフォームへと進化してきました。Mitaka 上で動くコンテンツが増えることはプラットフォームとしての Mitaka の価値を向上させることとなります。本連載

が、利用者のみなさんから新たなコンテンツが生み出される一助となれば幸いです。

## 謝 辞

加藤恒彦氏はじめ国立天文台 4D2U プロジェクトの皆様は長年にわたる Mitaka の開発・サポートに感謝いたします。Mitaka WG においては Mitaka の理解を深める様々な機会をいただき、ありがとうございました。また本連載に際しまして、「天文教育」編集委員の皆様、とりわけ松岡義一氏には文章の細部に至るまでご助言をいただき感謝いたします。

## 文 献

- [1] 加藤恒彦 (2019) 「天文ソフト「Mitaka」の最新機能 ～ユーザーによるカスタマイズ機能と教材開発への応用～」第 33 回天文教育研究会集録, pp157-160.
- [2] 河村聡人 (2023) 「Mitaka カスタマイズの手引き 1 ～保存と読み込み：日食（一地点）～」天文教育, vol.35, no.3, pp64-69.
- [3] 河村聡人 (2023) 「Mitaka カスタマイズの手引き (2) ～コマンドセット 1: 日食（多地点）～」天文教育, vol.35, no.4, pp28-33.
- [4] 河村聡人 (2023) 「Mitaka カスタマイズの手引き (3) ～コマンドセット 2: 星座～」天文教育, vol.35, no.5, pp23-28.
- [5] 河村聡人 (2023) 「Mitaka カスタマイズの手引き (4) ～シーケンス 1: 月の満ち欠け（宇宙）～」天文教育, vol.35, no.6, pp 2-7.
- [6] 河村聡人 (2023) 「Mitaka カスタマイズの手引き (5)」天文教育, vol.36, no.1,

pp15-21.

- [7] 波田野聡美 (2020) 「Mitaka による天文教育／普及ワーキンググループ活動報告」第 34 回天文教育研究会収録, pp92-93.
- [8] Mitaka WG の Google ドライブ  
<https://drive.google.com/drive/folders/1vPjyW-j1rEZ0AtGC8I5IrreJSL0eVlX-?usp=sharing>
- [9] Mitaka 同梱マニュアル：  
 ・ mitaka\_manual\_J.pdf  
 ・ features\_v174.pdf（バージョンに従い名前の一部は変更される）
- [10] Mitaka の利用上の注意  
<https://4d2u.nao.ac.jp/policy/>
- [11] Mitaka ダウンロードページ  
<https://4d2u.nao.ac.jp/mitaka/download/>
- [12] 河村聡人 (2021) 「Mitaka の改造のススメ」天文教育, Vol.33, No.5, pp50-52.
- [13] 河村聡人 (2023) 「探索型クイズゲーム『SPACE-CISTE』～Mitaka を使ったゲームの紹介と実践～」天文教育, Vol.35, No.3, pp70-73.
- [14] SPACE-CISTE 配布サイト  
<https://sites.google.com/view/mitaka-customize-package/カスタマイズパッケージ/space-ciste>



河村 聡人

### A.1 ソースコード

spaceciste\_sample.mcd

```
1 //初期化用コマンドセット
2 DefCommandSet[SpaceCiste_Init]
3 //問題文の字幕の初期化
```

```
4 SetSubtitle(WindowKey:T1,
5   Text: "メニューバーのコマンドから問題を選んでください",
6   Color:#10FFFF, Y:22.5)
7
8 //各問の正解表示の初期化
9 SetSubtitle(WindowKey:Ciste_Q1, Text:)
10 SetStrings(Ciste_Q2:)
11
12 //キーガイド字幕の表示
13 SetSubtitle(WindowKey:L1, Text:"メニューを開く/選ぶ:",
14   Color:#10FFFF, X:-16.75, Y:18.6)
15 SetSubtitle(WindowKey:L2, Text:"・Xキー", Color:#10FFFF, X:-19.5, Y:17.3)
16 SetSubtitle(WindowKey:L3, Text:"メニューを閉じる:",
17   Color:#10FFFF, X:-17.5, Y:14.9)
18 SetSubtitle(WindowKey:L4, Text:"・Zキー", Color:#10FFFF, X:-19.5, Y:13.6)
19
20 //最低限の表示設定
21 SwitchOn(4D2U_LOGO_DISP, ON_SCREEN_CREDITS_DISP, HEAD_LIGHT, SUBTITLES_DISP)
22 SetLandingMode(LANDING_MODE_SURFACE_VIEW)
23
24 DefCommandSet[SpaceCiste_1, MENU_BAR]
25   KeyString(第1問)
26   ExecCommandSet(SpaceCiste_Init)
27   SetSubtitle(WindowKey:T1, Text:"第1問 春はあけぼの…星は○○○。")
28   SetSubtitle(WindowKey:Ciste_Q1, Type:ObjectLabel, ObjectKey:HIP_17702,
29     Text: "「正」", Color:#10FFFF)
30
31 DefCommandSet[SpaceCiste_2, MENU_BAR]
32   KeyString(第2問)
33   ExecCommandSet(SpaceCiste_Init)
34   SetSubtitle(WindowKey:T1, Text:第2問 月にある大洋)
35   SetPlace(ObjectKey:SAT_E_MOON, PlaceKey:Ciste_Q2, Lat:18.4, Lon:-57.4,
36     R1:0.07, Color:#10FFFF, Size:2)
37   SetStrings(Ciste_Q2: "「解」")
38
39 //読み込み時実行コマンドセット
40 DefCommandSet[STATE]
41   ExecCommandSet(SpaceCiste_Init)
```